



SAILFISH OS



Разработка графического интерфейса приложения для Sailfish OS

Кирилл Чувилин, Открытая мобильная платформа

Содержание

1. Обзор интерфейса Sailfish OS
2. Принципы навигации в приложениях
3. Элементы интерфейса приложений
4. Рекомендации

Обзор интерфейса Sailfish OS

Основные принципы

Управление без сложностей

- Простые жесты
- Взаимодействие со всем экраном

Реальная многозадачность

Атмосферы управляют внешним видом

Приложения используют весь экран




Экран блокировки

Состояние устройства

- Время, дата, уведомления
- Другая информация о статусе

Провести слева или справа для разблокирования




Домашний экран

Открывается при разблокировании устройства или сворачивании приложения

Содержит «обложки» свёрнутых приложений

- Просмотр ключевой информации
- Управление основными функциями без необходимости разворачивания
- Разворот приложения на весь экран




Лента событий

Оповещения

- Уведомления от приложений
- Погода

Быстрый доступ

- Изменение настроек
- Некоторые действия




Сетка приложений

Содержит все доступные приложения

Можно упорядочивать и группировать иконки

Доступна отовсюду, если провести снизу экрана




Верхнее меню

Управление состоянием устройства

- Блокировка устройства
- Переключение бесшумный режим
- Смена атмосферы

Активируется, если провести сверху экрана




Атмосферы

Определяют внешний вид и поведение

- Фоновая картинка
- Цветовая схема
- Звуковая схема


Можно создавать свои из изображений



Принципы навигации в приложениях


Присоединённая страница: доступ к контенту

1. Нажать на категорию для просмотра содержимого
2. Провести от левого края экрана, чтобы вернуться к списку категорий



Присоединённая страница: доступ к действиям


1. Нажать на элемент для отображения доступных действий
2. Провести от левого края экрана, чтобы скрыть действия



Вытягиваемое меню

Интерактивное расширение страницы

- Встраивается сверху или снизу
- Изменение активного элемента по мере вытягивания



Диалоги

Подтверждение от пользователя

- Провести слева экрана для отмены
- Провести справа экрана для подтверждения



Элементы интерфейса приложений

Архитектура GUI-приложений

Декларативное описание UI с помощью QML

- Стандартный модуль QtQuick
- Специфичный модуль Silica
- «Легкая» логика на JavaScript

Запуск с помощью исполняемого кода на C++

- Основной файл QML встраивается в QQuickView
- Доступ к API устройства
- «Тяжелые» вычисления

Создать новое приложение

В Sailfish OS IDE

- Приложение из базового шаблона
Начало → Проекты → Новый проект → Sailfish OS →
SailfishOS Qt Quick Application
- Пример реализации элементов UI
Начало → Примеры → Sailfish Silica Component Gallery

Простейшая точка входа в приложение

Запускает `qml/{имя проекта}.qml`

```
#include <sailfishapp.h>
```

```
int main(int argc, char *argv[]) {  
    return SailfishApp::main(argc, argv);  
}
```

QML-модуль Silica

Модуль QML для приложений Sailfish OS

- Набор компонентов для построения UI (Sailfish Silica QML Types)

Использование уникальных элементов

- Вытягиваемое меню
- Обложки


Стандартный визуальный стиль для элементов

Поведение, ожидаемое от приложений Sailfish OS

Окно приложения: ApplicationWindow

```
import Sailfish.Silica 1.0

ApplicationWindow {
    id: applicationWindow
    allowedOrientations: Orientation.All
    _defaultPageOrientations: Orientation.All
    initialPage: Page { }
    cover: CoverBackground { }
}
```



Возможные ориентации устройства

`applicationWindow.orientation`:

- `Orientation.Portrait`
- `Orientation.Landscape`
- `Orientation.PortraitInverted` (не для смартфонов)
- `Orientation.LandscapeInverted`

Маски для проверки и настройки ориентаций

```
Orientation.PortraitMask =
```

```
    Orientation.Portrait | Orientation.LandscapeInverted
```

```
Orientation.LandscapeMask =
```


```
    Orientation.Landscape | Orientation.LandscapeInverted
```

```
Orientation.All =
```

```
    Orientation.PortraitMask | Orientation.LandscapeMask
```



Основа страницы приложения: Page

```
import QtQuick 2.2
import Sailfish.Silica 1.0
Page {
    SilicaFlickable {
        anchors.fill: parent
        contentHeight: column.height
        Column {
            id: column
            height: applicationWindow.height * 1.5
            PageHeader{ title: qsTr("Flickable Page") }
        }
        VerticalScrollDecorator { flickable: parent }
    }
}
```




Управление страницами: PageStack

```
import QtQuick 2.2
import Sailfish.Silica 1.0
Page {
    Column {
        width: parent.width
        spacing: Theme.paddingLarge
        PageHeader { title: qsTr("Page") + " " + pageStack.depth }
        Button {
            text: qsTr("Next page")
            anchors.horizontalCenter: parent.horizontalCenter
            onClicked: pageStack.push(Qt.resolvedUrl("StackPage.qml"))
        }
        Button {
            text: qsTr("Previous page")
            anchors.horizontalCenter: parent.horizontalCenter
            onClicked: pageStack.pop()
            enabled: pageStack.depth > 1
        }
    }
}
```




Список элементов: SilicaListView

```
import QtQuick 2.2
import Sailfish.Silica 1.0
Page {
    SilicaListView {
        header: PageHeader { title: qsTr("List Page") }
        anchors.fill: parent
        delegate: ListItem {
            Label {
                text: qsTr("Item") + " " + (model.index + 1)
                anchors.verticalCenter: parent.verticalCenter
                x: Theme.horizontalPageMargin
                color: highlighted ?
                    Theme.highlightColor : Theme.primaryColor
            }
        }
        model: 100
        VerticalScrollDecorator { flickable: parent }
    }
}
```




Галерея элементов: SilicaGridView

```
import QtQuick 2.2
import Sailfish.Silica 1.0
Page {
    SilicaGridView {
        header: PageHeader { title: qsTr("Grid Page") }
        anchors.fill: parent
        cellWidth: width / 4
        cellHeight: cellWidth
        delegate: BackgroundItem {
            width: 0.9 * GridView.view.cellWidth
            height: 0.9 * GridView.view.cellHeight
            Label {
                text: (model.index + 1)
                anchors.centerIn: parent
                color: highlighted ?
                    Theme.highlightColor : Theme.primaryColor
            }
        }
    }
    model: 100
}
```



Цикл слайдов: SlideshowView


```
import QtQuick 2.2
import Sailfish.Silica 1.0
Page {
    SlideshowView {
        id: slideshowView
        anchors.centerIn: parent
        width: Math.min(parent.width, parent.height)
        height: width
        delegate: BackgroundItem {
            width: slideshowView.itemWidth
            height: slideshowView.itemHeight
            Label {
                text: qsTr("Item") + " " + (index + 1)
                anchors.centerIn: parent
            }
        }
    }
    model: 5
}
```



Просмотр WEB: SilicaWebView

```
import QtQuick 2.2
import Sailfish.Silica 1.0
Page {
    Column {
        anchors.fill: parent
        SilicaWebView {
            id: webView
            width: parent.width
            height: parent.height - urlField.height
            url: "http://omprussia.ru"
        }
        TextField {
            id: urlField
            inputMethodHints: Qt.ImhUrlCharactersOnly
            text: "http://omprussia.ru"
            label: webView.title
            EnterKey.onClicked: {
                webView.url = text
                parent.focus = true
            }
        }
    }
}
```


Кирилл Чувиллин, Открытая мобильная платформа



SAILFISH OS


Ввод текста: TextField и TextArea

```
import QtQuick 2.2
import Sailfish.Silica 1.0
Page {
    Column {
        anchors.fill: parent
        PageHeader { title: qsTr("Text Input Page")}
        TextField {
            placeholderText: qsTr("Fill the field")
            label: qsTr("Text Field")
            width: parent.width
        }
        TextArea {
            placeholderText: qsTr("Fill the area")
            label: qsTr("Text area")
            width: parent.width
        }
    }
}
```































Обложка приложения: CoverBackground

```
import QtQuick 2.2
import Sailfish.Silica 1.0
CoverBackground {
    Label {
        id: label
        anchors.centerIn: parent
        text: applicationWindow.currentPageName
    }
    CoverActionList {
        CoverAction {
            iconSource:
                "image://theme/icon-cover-previous"
            onTriggered: pageStack.pop()
        }
    }
}
```




Иконки для CoverAction

Icon	Name	Icon	Name	Icon	Name
	icon-cover-alarm		icon-cover-answer		icon-cover-backup
	icon-cover-camera		icon-cover-cancel		icon-cover-dialer
	icon-cover-favorite		icon-cover-hangup		icon-cover-location
	icon-cover-message		icon-cover-mute		icon-cover-new
	icon-cover-next-song		icon-cover-next		icon-cover-pause
	icon-cover-people		icon-cover-play		icon-cover-previous-song
	icon-cover-previous		icon-cover-refresh		icon-cover-reject
	icon-cover-search		icon-cover-shuffle		icon-cover-subview
	icon-cover-sync		icon-cover-timer		icon-cover-transfers
	icon-cover-unmute				


Вытягиваемое меню: PullDownMenu

```
import Sailfish.Silica 1.0
Page {
    ListView {
        header: PageHeader {
            title: qsTr("Pull Down Menu Page")
        }
        anchors.fill: parent
        PullDownMenu {
            MenuItem {
                text: "Option"
                onClicked: console.log(qsTr("Option clicked"))
            }
            MenuLabel {
                text: qsTr("Informational label")
            }
        }
    }
}
```




Нижнее меню: PushUpMenu

```
import Sailfish.Silica 1.0
Page {
    ListView {
        header: PageHeader {
            title: qsTr("Push Up Menu Page")
        }
        anchors.fill: parent
        PushUpMenu {
            backgroundColor: "red"
            MenuItem {
                text: "Option"
                onClicked: console.log(qsTr("Option clicked"))
            }
            MenuLabel {
                text: qsTr("Informational label")
            }
        }
    }
}
```




Контекстное меню: ContextMenu

```
import QtQuick 2.2
import Sailfish.Silica 1.0
Page {
    SilicaListView {
        anchors.fill: parent
        model: 10
        delegate: ListItem {
            Label {
                id: label
                text: qsTr("Item") + (model.index + 1)
                anchors.centerIn: parent
            }
            menu: ContextMenu {
                MenuItem {
                    text: qsTr("Toggle bold font")
                    onClicked: label.font.bold = !label.font.bold
                }
            }
        }
    }
}
```




Подтверждение от пользователя: Dialog

```
import QtQuick 2.2
import Sailfish.Silica 1.0
Dialog {
    property string name
    Column {
        anchors.fill: parent
        DialogHeader { }
        TextField {
            id: nameField
            width: parent.width
            placeholderText: "What's your name?"
            label: "Name"
        }
    }
}
onDone: {
    if (result == DialogResult.Accepted) {
        name = nameField.text
    }
}
```




Подтверждение от пользователя: Dialog

```
import QtQuick 2.2
import Sailfish.Silica 1.0
Page {
    Column {
        anchors.fill: parent
        PageHeader { id: header }
        Button {
            text: qsTr("Set header")
            anchors.horizontalCenter: parent.horizontalCenter
            onClicked: {
                var dialog =
                    pageStack.push(
                        Qt.resolvedUrl("../dialogs/HeaderInputDialog.qml"),
                        {"name": header.title});
                dialog.accepted.connect(function() {
                    header.title = "New header: " + dialog.name;
                });
            }
        }
    }
}
```



Уведомления: Notification

```
import QtQuick 2.2
import Sailfish.Silica 1.0
import org.nemomobile.notifications 1.0
Page {
    Column {
        width: parent.width
        spacing: Theme.paddingLarge
        PageHeader { title: qsTr("Notification Page") }
        Button {
            anchors.horizontalCenter: parent.horizontalCenter
            Notification {
                id: notification
                category: "x-nemo.example"
                summary: qsTr("Notification summary")
                body: qsTr("Notification body")
            }
            text: qsTr("Application notification") +
                (notification.replacesId ?
                    " ID:" + notification.replacesId : "")
            onClicked: notification.publish()
        }
    }
}
```

























Иконки элементов интерфейса

source: "image://theme/..."

В документации: Sailfish Icon Reference

- Маленькие
- Средние
- Большие
- Действия обложки
- Уведомления

Icon	Name	Icon	Name	Icon	Name
	icon-lock-alarm		icon-lock-application-update		icon-lock-backup
	icon-lock-birthday		icon-lock-calendar		icon-lock-chat
	icon-lock-email		icon-lock-friend-request		icon-lock-information
	icon-lock-installed		icon-lock-invitation		icon-lock-locked
	icon-lock-missed-call		icon-lock-mms		icon-lock-more
	icon-lock-settings		icon-lock-sms		icon-lock-social
	icon-lock-system-update		icon-lock-transfer		icon-lock-voicemail
	icon-lock-warning				

<https://sailfishos.org/develop/docs/jolla-ambient/>

Рекомендации

Иконки приложений

- 16 допустимых форм
- Центральные или диагональные градиенты
- Разные размеры для разных экранов
- Шаблон:
<https://sailfishos.org/wp-content/uploads/2015/01/Sailfish-Apps-icon-template.zip>



Масштаб интерфейса: Theme.pixelRatio

```
import QtQuick 2.2
import Sailfish.Silica 1.0
Rectangle {
    color: "red"
    anchors.centerIn: parent
    width: 100 * Theme.pixelRatio
    height: Theme.itemSizeSmall
}
```


Выбор размера элементов: Screen.sizeCategory

```
import QtQuick 2.2
import Sailfish.Silica 1.0
Label {
    text: "Hello world!"
    anchors.centerIn: parent
    font.pixelSize: Screen.sizeCategory >= Screen.Large
                    ? Theme.fontSizeLarge
                    : Theme.fontSizeNormal
}
```


Выбор компоновки: Screen.sizeCategory

```
import QtQuick 2.2
import Sailfish.Silica 1.0
ApplicationWindow {
    initialPage: Screen.sizeCategory >= Screen.Large
                ? Qt.resolvedUrl("SplitViewPage.qml")
                : Qt.resolvedUrl("ListViewPage.qml")
}
```

Полезные обложки




Не правильно




Правильно

Статичные элементы: Theme.highlightColor



Не правильно



Правильно

Выбранные пункты: Theme.highlightColor




Не правильно




Правильно

Отступ сбоку: Theme.horizontalPageMargin




Не правильно




Правильно

Для нажатия не меньше Theme.itemSizeMedium




Не правильно




Правильно

Не более 4-х элементов в вытягиваемом меню



Не правильно




Правильно

Не показывать отключенное меню




Не правильно




Правильно

Показывать индикатор прокрутки




Не правильно




Правильно

Использовать действия вместо кнопок




Не правильно




Правильно

Использовать свойства placeholderText и label




Не правильно




Правильно

Использовать действия EnterKey



Не правильно



Правильно

Дополнительная информация



@SailfishRussia



merproject

- QML Applications
<http://doc.qt.io/qt-5/qmlapplications.html>
- Creating applications with Sailfish Silica
<https://sailfishos.org/develop/docs/silica/>
- Sailfish Silica Reference
<https://sailfishos.org/develop/docs/silica/sailfish-silica-all.html/>
- Sailfish Icon Reference
<https://sailfishos.org/develop/docs/jolla-ambient/>



Контакты

Кирилл Чувилин, Открытая мобильная платформа




k.chuvilin@omprussia.ru



[@kirill_chuvilin](https://twitter.com/kirill_chuvilin)



[kirill_chuvilin](https://vk.com/kirill_chuvilin)



**ОТКРЫТАЯ
МОБИЛЬНАЯ
ПЛАТФОРМА**